
$oninline
/*
 GAMS program used to estimate technical efficiency.

 Author:  John B. Walden
          NMFS/NEFSC
          166 Water St.
          Woods Hole, MA  02543
          (508) 495-2355
          John.Walden@Noaa.Gov

This version will estimate technical efficiency
based on an output oriented DEA model, with
constant returns to scale. Models can
be found in Fare, Grosskopf and Lovell (1994)

This version utilizes the GUSS solver in GAMS. 
For more information on the GUSS solver see the GAMS
web site www.gams.com*/

$OFFSYMLIST OFFSYMXREF OFFUELLIST OFFUELXREF

OPTION SOLPRINT=OFF, SYSOUT=OFF, LIMROW=0, LIMCOL=0;

/*NEXT DEFINE INPUTS AND OUTPUTS. There are 6 inputs x1,x2,x3,
x4,x5 and x6; there are 3 outputs y1,y2,y3. The data has 201 
observations*/

SET INOUT /x1*x6, y1*y3/

     OUTPUT(INOUT) /y1*y3/
     INPUT(INOUT) /x1*x6/

     OBS /1*201/
     SUBOBS(OBS) /1*201/
;

/*Next, define an alias for the set SUBOBS, and read in the 
data*/

alias (subobs, subobs1);

$OFFLISTING
TABLE ACT(OBS,INOUT)  INPUT OUTPUT TABLE
$ondelim
$INCLUDE "data2.csv"
$offdelim
$ONLISTING

parameter slice(inout) slice of input output data
          eff_k(obs) efficiency report;

VARIABLES

 theta      efficiency score



 weight(obs)  weights;

 POSITIVE Variable weight;

 EQUATIONS

   CONSTR1(output)  DEA constraint for each output
   CONSTR2(input)  DEA constraint for each input
;

   CONSTR1(output).. sum(subobs, weight(subobs)*ACT(subobs,output)) =G= Theta*slice(output);

   CONSTR2(input).. sum(subobs, weight(subobs)*ACT(subobs,input)) =L= slice(input);

MODEL TE /all/;

set h headers /modelstat, solvestat, objval/;
parameter scenrep(obs,h) solution report summary
           scopt /SkipBaseCase 1/;

set dict /subobs. scenario.''
          scopt. opt. scenrep
          slice. param. ACT
          theta. level. eff_k/;

slice(inout)=0; 

option lp=cbc;

solve te maximizing theta using lp scenario dict;

display scenrep, eff_k;

/*results are printed out to the file 'results_test.csv'*/

file sr /results_test.csv/;

sr.pc=5;
put sr;

put 'obs', 'theta';
put /; 
loop(subobs, put subobs.tl, scenrep(subobs,'objval'):4; 
  put /;

);
putclose;


